U. S. DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

MINCON, a BASIC program to control temperature and oxygen fugacity in furnaces

by

J. Stephen Huebner

Open-File Report 81-1324

This report is preliminary and has not been reviewed for conformity with U.S. Geological Survey editorial standards. Any use of trade names is for descriptive purposes only and does not imply endorsement by the USGS.

1981

CONTENTS

ILLUSTRATION

TABLES

Introduction

Following the pioneering studies of Van't Hoff (see Eugster, 1971) and Bowen (1928), experimental petrologists have simulated natural processes in the laboratory to understand the conditions at which minerals and rocks crystallize. For half a century, these studies have focused upon the determination of equilibrium conditions. Charges (samples) of known bulk composition were subjected to steady values of the intensive parameters temperature, pressure, and fugacities of volatile components; the charges were quenched to ambient conditions and then examined. Run durations used were generally believed sufficient for the charge to achieve an equilibrium state. This research enabled petrologists to comprehend many aspects of magma generation and crystallization by equilibrium processes, but it did not permit satisfactory explanations of processes that never attained equilibrium or that proceeded by a succession of local equilibrium states. Such processes must be understood if petrologists are to interpret correctly the distinct textures and the intercrystalline minor element distributions in igneous and volcanic rocks or to measure cooling rates from diffusively zoned crystals or microstructures.

Recent efforts in experimental mineralogy and petrology have stressed the departure from equilibrium and thus have necessitated the control of temperature as a function of elapsed time. Such efforts include dynamic crystallization studies of lunar basalts (Donaldson et al., 1975; Thornber and Huebner, 1980) and my own attempts to measure Fe-Mg diffusion rates in silicate minerals (unpublished). The perfection with which these studies model natural processes is limited, in part, by the difficulty of manually controlling more than one of the important intensive parameters, especially for the long periods of time characteristic of even relatively fast petrologic processes.

Manually controlled furnaces are not attended 24 hours per day, and researchers have difficulty in making the precise adjustments in CO_2 and H_2 gas flow rates necessary to vary furnace f_{O_2} as a smooth function of temperature or time. In this report, I describe a program that will cause a laboratory process controller to monitor and control gas mixing furnaces for experiments of long (indefinite) duration. For diffusion and dynamic crystallization studies, temperature, oxygen fugacity, and time are the critical variables, but the program can be modified to control other variables.

The program MINCON was written to cause an Analog Devices Macsym II process controller to satisfy several requirements:

1. Monitor the temperature and oxygen fugacity of conventional gas-mixing and quenching furnaces equipped with zirconia cells.

2. Control the temperature and oxygen fugacity of such furnaces as a function of time and temperature, respectively.

3. Without disturbing other active processes (experiments), permit any experiment to be terminated, modified, and started again.

4. In the event that electrical power is interrupted and restored, automatically resume the control program as if the power had not failed. This feature permits experiments of unusually long duration.

Laboratory Apparatus

The process controller is connected to laboratory equipment such that it can react to or interact with five quenching furnaces and one high-pressure, piston-cylinder apparatus (Fig. 1). Each furnace and the piston-cylinder apparatus are provided with a $Pt\text{-}Pt_{90}Rh_{10}$ thermocouple for measuring temperature. Temperatures are regulated by power controllers; four of the power controllers accept a remote 0-5 VDC (volts of direct current) analogue temperature control signal. In three of the furnaces, a flow of CO_2 + H_2 (or CO) gas maintains the furnace atmosphere at a desired oxygen fugacity (Nafziger et al., 1971) and an electrochemical sensor measures the furnace oxygen fugacity (Sato, 1971; Huebner, 1975). The flow rates of CO_2 and H_2 (or CO) to one furnace are adjusted manually; the CO_2 and H_2 (or CO) flow rates to the other two controlled-atmosphere furnaces are individually controlled by electronic mass-flow controllers where flow rate is proportional to a 0-5 VDC control signal.

Description of Program

MINCON is written in the enhanced BASIC language, revision 3.41, provided by Analog Devices, Inc. When the compiled version alone is in memory, the program will run in 32K words of memory. Appendix I is a printout of the program. The program sets up seven tasks. The main task (1) defines six subsidiary tasks and their periodicity, is responsible for restarting the program in case of electrical power failure, and can be interrupted by keyboard command to examine or change the values of variables and constants and to stop or restart any of the six subsidiary tasks. Each task can be visualized as functioning as a discrete program with its own pointer. The various tasks can be distinguished by their statement numbers: task 1 has statements 10 to 999; task 3 has 3000-3999, etc. The six subsidiary tasks or programs perform the following functions:

Tasks 3, 5) Measure temperature, oxygen fugacity, and elapsed time; control temperature and gas flow rates of CO_2 and H_2 for each of two gas-flow furnaces.

Task 4) Measures and controls temperature; measures elapsed time and fO_2 for one gas-flow furnace.

Task 6) Measures temperature, pressure, and elapsed time of 2 furnaces and piston-cylinder apparatus; controls temperatures of piston-cylinder apparatus.

Task 7) Formats the video output to the CRT (cathode ray tube), controls the cursor and type size, and keeps program time.

Task 8) Formats the printed output and controls the printer.

Explanation of two tasks (1,3) should suffice to make clear the operation of the program. Task 1 is unique; tasks 4-6 embody operations included in task 3, and tasks 7 and 8 use conventional BASIC format and print statements to put results out to a printer and the video screens.

At the beginning of the program, task 1 prints title and date, reserves space, and assigns initial values to various "constants" used subsequently (Table 1). These "constants" are placed in the main task so that the initial value may be adapted by subsequent operators or programmers to the needs of individual experiments. Task 1 then defines the six secondary tasks (lines 300-800); these statements include the line number at which each task begins and a priority (100 to 200 in the program). If statements from two active tasks require the CPU (central processing unit) at the same time, the statement from the task having higher priority will be executed first.

Lines 903-971 establish the periodicity with which individual secondary tasks are executed and start execution of these tasks. Task 7 executes most frequently (30-second intervals); task 3 is executed every 85 seconds. Task 8 controls printer output; measurements are printed every 300, 3600, or 7200 seconds, depending upon the value of J8 (970, 960, or 920, respectively) at the time Task 8 was last activated.

After the tasks are started, the task 1 pointer moves into lines 986 to 993, a loop designed to keep the main task active. Exit from the loop is provided by a value of J8 other than 989 or by an automatic restart. If the automatic restart statement (line 986) is activated, control is passed to line 994, the time of the power failure is printed, the subsidiary tasks are restarted (line 997), then control is passed back to the main loop beginning with line 986. In either case, departure from the main loop is only momentary.

Task 3, lines 3000 to 3999, first measures the thermocouple (microvolts) and oxygen probe (volts), ten times each. The average thermocouple microvoltage reading (line 3100) is converted to °C (line 3110) and 1/°K (line 3120) by means of a polynomial of the form $°C = a + bx + cx^2 + dx^3 + ex^4$ where x is microvolts. The average oxygen probe EMF reading (also line 3100) is used to compute the log fO_2 value of the furnace atmosphere, F3 (line 3160), using the Nernst equation, $EMF = (RT/4F) \cdot \ln (fO_2 \text{ furnace}/fO_2 \text{ reference})$ where the reference oxygen fugacity is 1 atmosphere. The run or elapsed time T3 in line 3190 is simply the difference between the current system time (T9) and the system time at the initiation of an experiment (Q3).

The desired temperature (D3) is computed as a function of elapsed time (T3) in lines 3400-3410, according to the equation.

$$T = T_{max} - \frac{T_{max} - T_{min}}{a + b\left[\frac{k}{T3}\right] + c\left[\frac{k}{T3}\right]^2 + d\left[\frac{k}{T3}\right]^3 + e\left[\frac{k}{T3}\right]^4 + f\left[\frac{k}{T3}\right]^5}$$

This equation reproduces the form of conductive heat-loss cooling curves calculated by use of the error function (see Sanford and Huebner, 1980 for examples of the curve modelled by this equation). The temperature range and time can be scaled by adjusting T_{max}, T_{min}, and k [A3(1), A3(2), and A3(9) in the program, line 133]. The difference between the desired and measured temperatures (line 3420) is used to increment the temperature setpoint voltage (lines 3450, 3470).

The desired oxygen fugacity (O3) is calculated as a function of reciprocal measured temperature (K3) by use of a polynomial

$$O3 = a + b\,(K3) + c\,(K3)^2 + d\,(K3)^3 + e\,(K3)^4 + f\,(K3)^5$$

in line 3500. The oxygen fugacity-temperature relationships of standard oxygen buffer curves (Huebner, 1971) are approximately linear and can be calculated from only the first two terms. Additional terms can account for departures from linearity. Desired and measured oxygen fugacity values are compared; then the difference is used to calculate new CO_2 and H_2 flow rates and new values of the gas-flow valve control-voltages (lines 3600-3720). For reference in the output, the difference in oxygen fugacity between the iron-wüstite assemblage (Huebner, 1971) and the furnace gas is calculated (lines 3810, 3998). The DISMISS statement (line 3998) keeps the task alive for periodic reactivation.

Time Constants

The control of quenching-furnace temperature and oxygen fugacity is inherently slow. Owing to the relatively slow rates of heat transport and gas flow, several seconds to minutes elapse before the result (changed value of process variable) of a changed control signal is sensed in the furnace. If time constants in the program are not matched to time constants inherent in the laboratory apparatus, the controlled temperature and fO_2 will oscillate or follow a step function, rather than smooth temperature-time and fO_2-time curves.

MINCON permits the operator to adjust programmed constants to match the time constants inherent in individual laboratory furnaces by modifying program statements or values of variables. The program statements control the periodicity with which a task is executed. Values of variables determine the relationship (gain) between the error signal and the response. The gain for the gas-flow control voltage is a variable, to be determined by the operator, because the relationship between change in mixing ratio and change in oxygen fugacity is critically dependent upon bulk composition, in addition to the time constant due to gas-flow rates in the control loop. At relatively high fO_2 values, the gain should be small (Ei=1), whereas at low fO_2 values, the gain should be high (Ei=5). Too large a gain will result in oscillations in fO_2. Too small a gain (or too long a periodicity) results in a stepfunction with time. Similarly, if the gas composition is to change rapidly with time, as during rapid cooling, a relatively large value of the gain is appropriate. Temperature setpoint gain is also a variable because the gain value is a unique property of the furnace, being due in part to the position of the thermocouples within the furnace. Satisfactory gain values were obtained by trial and error after adjusting the temperature controllers for stability in local (constant temperature) mode. These values are listed in Table 2.

Example of Operation and Printout

A compiled version of MINCON must be loaded into memory from tape file # (see Appendix II):

NEW RTN
LOAD MTØ:# RTN

If the program was previously listed onto the tape file, use

NEW RTN*
COMPILE MTØ:# RTN

Bring the furnaces to the desired initial temperature and oxygen fugacity by using the manual control (local) mode. When the intensive parameters in the furnaces are close to the desired values (how close will depend upon the tolerable amount of mismatch when automatic or remote mode is initiated), type RUN RTN to start the program (see Appendix II). Within several seconds of the warning bell, default values of the setpoints will be put out by the controller. At this time, the operator may switch any furnace to automatic (remote or program) control or adjust initial control variables [Vi, Gi, Hi, Ei, Ai(1), Ai(2), and Ai(9) are likely to need changing; see Table 1]. To adjust the variables, first type CTRL/A.** When the control variables are adjusted satisfactorily (remember, control voltages are changed only every ·90 seconds by the program), a task may be halted at steady state (SUSPEND i), and the output values of the variables will remain constant.

As the experiment is begun, set Qi equal to the current value of the program time T9 (thereby initializing the run-time to zero) and type ALERT i RTN (see Appendix III). The process controller will begin to control temperature and fO_2, attempting to maintain actual values equal to the calculated values. At this time it is advisable to resume the main task (CON Rtn) so that the power fail-automatic restart feature, which is controlled by task 1, is operative. In the example shown in Appendix III, furnace 3 has been brought to temperature and an oxygen fugacity close to that desired (the value of the iron-wüstite buffer). Following the entry of ALERT 3, task 3 assumed control of both temperature and oxygen fugacity. Within 4 minutes, furnace fO_2 was adjusted satisfactorily; the value of delta log (fO_2 in atm) is close to zero in the report printed 2 hours later.

Note the use of the CON to reactivate the main task in Appendix III, enabling the automatic restart command sequence. Subsequent entry of CTRL/A halts the main task and again permits the operator to stop or start tasks and to examine or change variables (see Table 3).

*Underscore indicates that the set of characters is a single keystroke.
**Slash indicates that both keys are depressed simultaneously.

Two notes of caution are necessary. When zeroing or calibrating a peripheral analogue device such as an electrometer, strip chart recorder, or digital thermometer, the analogue input to the process controller may be disturbed. The controller will sense the perturbed input value and attempt to correct this change by altering its control (output) signals. This problem can be avoided by suspending the appropriate task (SUSPEND i) during the zeroing or calibrating procedure, then subsequently resuming the task (ALERT i). During suspension, a task's output signals are maintained constant and its computing is halted. When resumed, the time will be updated and new measurements (and output) will be made. The second warning concerns the automatic restart feature, which is controlled by Task 1. When the main task is halted, either by use of CTRL/A or by BASIC program error, the automatic restart feature is disabled. In this case, a power interuption will halt <u>all</u> tasks and set all outputs to 0 volts; however, the program and values of all variables remain stored in memory. In the event that the main task halts, print the values of the variables, restart the program with RUN, then replace the default values of the variables with those to be used.

Acknowledgements

Gilpin R. ("Rob") Robinson of the U.S. Geological Survey (USGS) wrote and kindly made available the equation that reproduces the cooling curves originally derived by use of the erf function. Use of this simple equation conserves considerable memory within the computer. Robinson and Priestley Toulmin III, also of the USGS, and John Bowles of Analog Devices, Inc., reviewed this report and offered helpful suggestions.

References Cited

Bowen, N.L. (1928) The evolution of the igneous rocks. Princeton Univ. Press, Princeton, N.J. x+ 334 pages.

Donaldson, C.H., T.M. Usselman, R.J. Williams, and G.E. Lofgren (1975) Experimental modeling of the cooling history of Apollo 12 olivine basalts. Proc. Lunar Sci. Conf. 6th, 843-869.

Eugster, H.P. (1971) The beginnings of experimental petrology. Science, 173, 481-489.

Huebner, J.S. (1971) Buffering techniques for hydrostatic systems at elevated pressure, in Research techniques for high pressure and high temperature, G.C. Ulmer, editor, p. 123-177. New York: Springer-Verlag.

Huebner, J.S. (1975) Oxygen fugacity values of furnace gas mixtures. Amer. Mineral. 60, 815-823.

Nafziger, R.H., G.C. Ulmer, and E. Woermann (1971) Gaseous buffering for the control of oxygen fugacity at one atmosphere, in Research techniques for high pressure and high temperature, G.C. Ulmer, editor, p. 9-41. New York: Springer-Verlag.

Sanford, R.F. and J.S. Huebner (1980) Model thermal history of 77115 and implications for the origin of fragment-laden basalts. Proc. Conf. Lunar Highlands Crust, 253-269.

Sato, M. (1971) Electrochemical measurements and control of oxygen fugacity and other gaseous fugacities with solid electrolyte sensors, in Research techniques for high pressure and high temperature, G.C. Ulmer, editor, p. 43-99. New York: Springer-Verlag.

Thornber, C.R. and J.S. Huebner (1980) An experimental study of the thermal history of fragment-laden "basalt" 77115. Proc. Conf. Lunar Highlands Crust, 233-252.

Table 1. Keyboard Commands

RUN - Begin entire program

CTRL/B - Halt entire program, set output voltages to zero

CTRL/A - Interrupt main task (1); other tasks continue execution

Keyboard Commands recognized only after CTRL/A

CON - Resumes execution of main task

PRINT X - Returns value of variable X

X=N - Assigns X a new value N

SUSPEND n - Interrupts execution of task n

ALERT n - Resumes execution of task n

Values ("constants") the operator will need to change

A3 - A6 - gain (change in control signal/error signal), temperature setpoint

J8 - Determines frequency of printout (J8 = 920 for every 2 hours, default; 960 for every 60 minutes; 970 for every 5 minutes). Must be followed by RTN CON RTN

A3(i) - A6(i) Cooling curve constants for tasks 3 to 6

O3(i), O5(i) - Polynomial equation constants to control fO_2

A9(i) - Thermocouple conversion constants, used to convert microvolts to degrees C.

Q1 to Q6 - Values of system time in hours (T9) at the initiation of particular experiments in Furnaces 1 to 5 and the piston-cylinder apparatus (Q6).

V3 to V6 - Values of the programmable temperature setpoint at the initiation of an experiment, in volts.

E1 to E6 - gain (change in control signal/error signal), gas-flow controllers

B1 to B6 - Temperature correction, in °C

L8 - Pressure transducer conversion factor, to yield kilobars

G3, G5 - Initial CO_2 flow rates, SCCM (standard cubic centimeters per minute)

H3, H5 - Initial H_2 flow rates, SCCM

S3, S5 - Total gas-flow rate to be maintained during an experiment

Table 2: Values of Time Constants

Variable	Task periodicity	Gain
temperature	90 seconds	0.001-0.002
fO_2 (at iron-wüstite)	90 seconds	1-5
fO_2 (above iron-wüstite)	90 seconds	1-3

Table 3. List of Internal Variables Used by MINCON

i designates task number

Ji	counter in FOR NEXT loop
Zi	analogue voltage measurement, thermocouple
Yi	running sum of Zi in FOR NEXT loop
Xi	average of 10 voltage measurements, thermocouple
Ci	measured temperature, °C
Ti	elapsed time since initiation of experiment
T9	elapsed system time since program initiated (by RUN command)
Qi	value of system time when experiment initiated
Li	analogue voltage measurement, oxygen cell
Pi	average of 10 voltage measurements, oxygen cell
Ki	reciprocal of absolute temperature
Ni	running sum of Li in FOR NEXT loop
Fi	log of measured oxygen fugacity in units of log bars
Ui	part of equation used to calculate Di
Mi	part of equation used to calculate Di
Di	desired temperature
Wi	difference between desired and measured temperature
Vi	temperature setpoint, volts
Oi	desired oxygen fugacity
Gi	CO_2 flow rate SCCM
Hi	H_2 flow rate SCCM
Si	total gas flow rate, SCCM
Ii	calculated log fO_2 of iron-wüstite assemblage at Ci
Ri	difference between Ii (calculated) and Fi (measured log fO_2)
L7	pressure, kilobars

Table 3. List of internal variables - continued

L9	constant relating number of seconds measured by the system's internal stopwatch to hours of elapsed time. Nominally 3600.
W8	difference in log fO_2, calculated - measured values, Furnace 3
W9	difference in log fO_2, calculated - measured values, Furnace 5

Figure 1. Schematic diagram showing relationship of process controller to laboratory furnaces and piston-cylinder apparatus. The small numbers in the upper right hand corner of each box are program task numbers, here associated with the system components the various tasks control. The five furnaces and piston cylinder apparatus all include thermocouples for furnace power controllers. Furnaces 3, 4, and 5 contain electrochemical (oxygen fugacity) cells; Furnaces 3 and 5 have remotely controlled gas flow control values; and the piston cylinder apparatus contains a pressure transducer. VDC is volts, direct current.


8
Printer
7
Video Screen
6
Furn #2
"Sun-dance"
6
Furn #1
"Butch"
1,7
Macsym - II process controller
Analogue to digital converter
Input and Output cards
Digital to analogue converter
Thermocouple
0–15 mV
Thermocouple
0–15 mV
5
Furnace #5
H2 control
0-5 V DC
CO2 control
0-5 V DC
Oxygen fugacity
0-2 V DC
Temp setpoint
0-5 V DC
Thermocouple
0-15 mV DC
4
Furnace #4
Oxygen fugacity
0-2 V DC
Temp setpoint
0-5 V DC
Thermocouple
0-15 mV DC
Same as Furnace #5
3
Furnace #3
6
Piston-cylinder apparatus
Thermocouple
0–15 mV DC
Temp setpoint
0–5 V DC
Pressure Transducer
0–1 V DC

APPENDIX I. Printout of MINCON

```
 10 T9=0 J8=920 L9=3600
 90 DIM N$(31),L$(3),A$(8),B$(8),C$(10),W$(28),D$(10)
100 N$="MINCON VERSION OF JULY 24, 1981"
105 OPENW :2"$QTO:0"
110 PNT 15 PRINT :2 N$ PRINT :2
113 PRINT :2 "PROGRAM ACTIVATED:  ";PTIME :2 PRINT :2 PRINT :2 PRINT :2
115 L$="--#"
118 PRINT :2 "                 ";
119 PRINT USING:2 "     FURNACE"L$,1,2,3,4,5,6 PRINT :2 PRINT :2
120 CLOSE :2
130 DIM A3(9),A4(9),A5(9),A6(9) @COOLING CURVE CONSTANTS
133 A3(1)=1230 A3(2)=1045 A3(3)=1.02714 A3(4)=3.11849 A3(5)=-1.51518 A3(6)=1.69106 A3(7)=-0.39303 A3(8)=0.05308 A3(9)=1.00000
134 A4(1)=1230 A4(2)=1045 A4(3)=1.02714 A4(4)=3.11849 A4(5)=-1.51518 A4(6)=1.69106 A4(7)=-0.38303 A4(8)=0.05308 A4(9)=1.00000
135 A5(1)=1230 A5(2)=1045 A5(3)=1.02714 A5(4)=3.11849 A5(5)=-1.51518 A5(6)=1.69106 A5(7)=-0.39303 A5(8)=0.05308 A5(9)=1.00000
136 A6(1)=1230 A6(2)=1045 A6(3)=1.02714 A6(4)=3.11849 A6(5)=-1.51518 A6(6)=1.69106 A6(7)=-0.39303 A6(8)=0.05308 A6(9)=1.00000
138 A3=.001 A4=.001 A5=.001 A6=.001 @DEFAULT VALUES OF GAIN, TEMPERATURE CONTROL (IN VOLTS PER DEGREE ERROR)
140 DIM O3(6),O4(6),O5(6) @CONSTANTS PGM F O2
143 O3(1)=6.57 O3(2)=-27215 O3(3)=0 O3(4)=0 O3(5)=0 O3(6)=0
145 O5(1)=6.57 O5(2)=-27215 O5(3)=0 O5(4)=0 O5(5)=0 O5(6)=0
190 DIM A9(5) @CONVERSION CONSTANTS FOR PT-PT90.RH10 THERMOCOUPLES
191 A9(1)=4.16705E+1 A9(2)=1.15434E-1 A9(3)=-1.67828E-6 A9(4)=-1.08458E-11 A9(5)=1.83797E-15
210 G1=0 G2=0 G3=0 G4=0 G5=0 G6=0 @SYSTEM TIME AT INITIATION OF EXPERIMENTS
220 V3=3.78 V4=3.68 V5=3.77 V6=0 @INITIAL TEMPERATURE SETPOINTS IN VOLTS
226 B1=8.5 B2=9.2 B3=25.8 B4=12.8 B5=23.0 B6=0
227 L8=1.0138 @PRESSURE TRANSDUCER FACTOR
233 G3=35 H3=55 S3=90 E3=5 @INITIAL CO2, H2, AND TOTAL GAS FLOW RATES IN SCCM; RATE OFCHANGE
235 G5=72 H5=18 S5=90 E5=3
300 TASK 3,3000,220 @FURN 3
400 TASK 4,4000,230 @FURN 4
500 TASK 5,5000,240 @FURN 5
600 TASK 6,6000,150 @PISTON-CYLINDER
700 TASK 7,7000,250 @CONTROLS CRT VIDEO DISPLAY
800 TASK 8,8000,100 @PRINTOUT
903 ACTIVATE 3 PER 85
904 ACTIVATE 4 PER 90
905 ACTIVATE 5 PER 95
906 ACTIVATE 6 PER 60
907 ACTIVATE 7 PER 30
920 ACTIVATE 8 PER 7200
921 GOTO 988
960 ACTIVATE 8 PER 3600
961 GOTO 988
970 ACTIVATE 8 PER 300
971 GOTO 988
986 RESTART MAIN,994
987 JUMP J8
988 J8=989
989 RESTART 3,3000 RESTART 4,4000 RESTART 5,5000 RESTART 6,6000 RESTART 7,7000 RESTART 8,8000
992 WAIT 1
993 GOTO 986
994 OPENW :2"$QTO:0"
995 PRINT :2 PRINT :2"POWER FAILED AT ";PTIME :2 PRINT :2 PRINT :2 PRINT :2
996 CLOSE :2
997 RESUME 3 RESUME 4 RESUME 5 RESUME 6 RESUME 7 RESUME 8
998 PNT 15 @CHANGES TO 64 CHARACTER LINE ON VIDEO SCREEN
999 GOTO 986
```

Appendix I. continued

```
3000 Y3=0 Z3=0 L3=0 N3=0 @TEND FURNACE #3
3030 FOR J3=1 TO 10
3040   Z3=AIN(0,4,9)*1E6 @MEASURE TC MICROVOLTS
3050   Y3=Y3+Z3
3060   L3=AIN(0,1,3) @MEAS OXYGEN PROBE VOLTS
3070   N3=N3+L3
3080   WAIT .22
3090 NEXT
3100 X3=Y3/10 P3=N3/10
3110 C3=POLY(X3,A9(1),5)+B3 @FIND TEMP CELSIUS
3120 K3=1/(C3+273.2) @CALCULATES 1/K FOR INTERNAL USE
3160 F3=-20160*P3*K3 @FIND LOG OXYGEN FUGACITY IN FURNACE
3180 T3=T9-Q3+.001 @STORE ELAPSED TIME OF EXPT IN HOURS
3400 U3=A3(3)+A3(4)*(A3(9)/T3)+A3(5)*(A3(9)/T3)^2
3405 M3=A3(6)*(A3(9)/T3)^3+A3(7)*(A3(9)/T3)^4+A3(8)*(A3(9)/T3)^5
3410 D3=A3(1)-((A3(1)-A3(2))/(U3+M3)) @CALC DESIRED TEMP AT TIME T3
3420 W3=D3-C3 @CALCULATES DIFFERENCE BETWEEN DESIRED AND MEASURED TEMPERATURE
3450 V3=V3+W3*A3 @CALC NEW TEMPERATURE SETPOINT IN VOLTS
3470 AOT(2,0)=V3 @PUTS OUT TEMP SETPOINT V3 IN VOLTS
3500 O3=POLY(K3,O3(1),6) @CALC DESIRED FO2 AT TEMP C3
3600 W8=O3-F3
3630 G3=G3+W8*E3
3635 H3=S3-G3
3710 AOT(1,0)=G3/60 @PUTS OUT CO2 CONTROL VALVE VOLTAGE
3720 AOT(1,2)=H3/20 @PUTS OUT H2 CONTROL VALVE VOLTAGE
3800 I3=6.57-27215*K3
3810 R3=F3-I3
3998 DISMISS
3999 GOTO 3000 @------------------------------------------------------------------------------------------
4000 Y4=0 Z4=0 L4=0 N4=0 @TEND FURNACE #4
4030 FOR J4=1 TO 10
4040   Z4=AIN(0,5,9)*1E6 @MEAS TC MICROVOLTS
4050   Y4=Y4+Z4
4060   L4=AIN(0,2,3) @MEAS OXYGEN PROBE VOLTS
4070   N4=N4+L4
4080   WAIT .22
4090 NEXT
4100 X4=Y4/10 P4=N4/10
4110 C4=POLY(X4,A9(1),5)+B4 @FIND TEMP CELSIUS
4120 K4=1/(C4+273.2)
4160 F4=-20160*P4*K4 @FIND OXYGEN FUGACITY IN FURNACE #4
4180 T4=T9-Q4+.001 @STORE ELAPSED TIME OF EXPT IN HOURS
4400 U4=A4(3)+A4(4)*(A4(9)/T4)+A4(5)*(A4(9)/T4)^2
4405 M4=A4(6)*(A4(9)/T4)^3+A4(7)*(A4(9)/T4)^4+A4(8)*(A4(9)/T4)^5
4410 D4=A4(1)-((A4(1)-A4(2))/(U4+M4)) @CALC DESIRED TEMP AT TIME T4
4420 W4=D4-C4
4450 V4=V4+W4*A4
4470 AOT(2,1)=V4 @PUTS OUT TEMP SETPOINT V4 IN VOLTS
4800 I4=6.57-27215*K4 @CALC FO2 OF IRON-WUSTITE AT MEASURED TEMPERATURE C4
4810 R4=F4-I4
4998 DISMISS
4989 GOTO 4000 @------------------------------------------------------------------------------------------
5000 Y5=0 Z5=0 L5=0 N5=0 @TEND FURNACE #5
5030 FOR J5=1 TO 10
5040   Z5=AIN(0,6,9)*1E6 @MEASURE TC MICROVOLTS
5050   Y5=Y5+Z5
5060   L5=AIN(0,3,3) @MEAS OXYGEN PROBE VOLTS
5070   N5=N5+L5
5080   WAIT .22
5090 NEXT
5100 X5=Y5/10 P5=N5/10
```

Appendix I. continued

```
5110 C5=POLY(X5,A9(1),5)+B5 @ FIND TEMP CELSIUS
5120 K5=1/(C5+273.2)
5180 F5=-20160*P5*K5 @FIND OXYGEN FUGACITY IN FURNACE #5
5190 T5=T9-G5+.001 @STORE ELAPSED TIME OF EXPT IN HOURS
5400 U5=A5(3)+A5(4)*(A5(9)/T5)+A5(5)*(A5(9)/T5)^2
5405 M5=A5(6)*(A5(9)/T5)^3+A5(7)*(A5(9)/T5)^4+A5(8)*(A5(9)/T5)^5
5410 D5=A5(1)-((A5(1)-A5(2))/(U5+M5)) @CALC DESIRED TEMP AT TIME T5
5420 W5=D5-C5
5450 V5=V5+W5*A5
5470 AOT(2,2)=V5 @PUTS OUT TEMP SETPOINT V5 IN VOLTS
5500 O5=POLY(K5,O5(1),6) @CALC DESIRED FO2 AT TEMP C5
5600 W9=O5-F5
5630 G5=G5+W9*E5
5635 H5=S5-G5
5710 AOT(1,1)=G5/40 @PUTS OUT CO2 VALVE CONTROL VOLTAGE
5720 AOT(1,3)=H5/10 @PUTS OUT H2 VALVE CONTROL VOLTAGE
5800 I5=6.57-27215*K5 @CALC FO2 OF IRON-WUSTITE AT MEAS TEMP
5810 R5=F5-I5
5998 DISMISS
5999 GOTO 5000 @------------------------------------------------------------------------------------------
6000 Y6=0 Z6=0 Y7=0 Z7=0 @TEND PISTON CYLINDER
6030 FOR J6=1 TO 10
6040   Z6=AIN(0,15,9)*1E6 @MEAS TC MV
6041   WAIT 0.21
6043   Z7=AIN(0,9,4) @MEAS PRESSURE TRANSDUCER VOLTS
6050   Y6=Y6+Z6
6051   Y7=Y7+Z7
6060 NEXT
6065 X6=Y6/10
6070 C6=POLY(X6,A9(1),5) @FIND TEMP CELSIUS
6080 L7=Y7*L6 @CONVERTS SUM OF TRANSDUCER MEASUREMENTS TO KILOBARS
6100 T6=T9-G6+.001 @STORE ELAPSED TIME OF EXPT IN HOURS
6400 U6=A6(3)+A6(4)*(A6(9)/T6)+A6(5)*(A6(9)/T6)^2
6405 M6=A6(6)*(A6(9)/T6)^3+A6(7)*(A6(9)/T6)^4+A6(8)*(A6(9)/T6)^5
6410 D6=A6(1)-((A6(1)-A6(2))/(U6+M6)) @CALC DESIRED TEMP AT TIME T6
6420 W6=D6-C6
6450 V6=V6+W6*A6
6470 AOT(2,3)=V6 @PUTS OUT TEMP SETPOINT V6 IN VOLTS
6500 Y1=0 Y2=0 Z1=0 Z2=0 @MONITOR FURNACES 1,2 (BUTCH AND SUNDANCE)
6530 FOR J1=1 TO 10
6532   Z1=AIN(3,0)*1E6 @MEAS TC MICROVOLTS, BUTCH
6534   Y1=Y1+Z1
6542   Z2=AIN(3,2)*1E6 @MEAS TC MICROVOLTS, SUNDANCE
6544   Y2=Y2+Z2
6546   WAIT 0.31
6548 NEXT
6565 X1=Y1/10 X2=Y2/10
6570 C1=POLY(X1,A9(1),5)+B1 @FIND TEMP CELSIUS OF BUTCH
6571 C2=POLY(X2,A9(1),5)+B2 @FIND TEMP CELSIUS OF SUNDANCE
6590 T1=T9-G1 T2=T9-G2 @STORE ELAPSED TIME OF EXPERIMENTS IN HOURS
6998 DISMISS
6999 GOTO 6000
7000 WAIT 20 @PAUSE TO PERMIT VARIABLES TO BE ASSIGNED VALUES ---------------------------------------------
7001 PNT 7 @SOUNDS WARNING BELL
7002 WAIT 5
7005 PNT 5 @CLEARS CRT SCREEN IN PREPARATION FOR NEW DISPLAY CYCLE
7010 PRINT PTIME PRINT
7020 PRINT "PGM TIME =";PRINT T9;PRINT "HRS"
7040 PRINT "FURNACE NO.      1       2       3       4       5      6"
7041 PRINT
7045 A$="--####.#"
7047 PRINT "RUN TIME HRS";
7050 PRINT USING A$,T1,T2,T3,T4,T5,T6
```

Appendix I. continued

```
7051 PRINT
7055 B$="--####.#"
7057 PRINT "TEMP CELSIUS";
7060 PRINT USING B$,C1,C2,C3,C4,C5,C6
7061 PRINT
7062 C$="--###.##"
7063 PRINT "DELTA TEMP CELSIUS          ";
7064 PRINT USING C$,W3,W4,W5
7065 PRINT
7067 PRINT "LOG ATM FO2                 ";
7070 PRINT USING C$,F3,F4,F5
7071 PRINT
7075 D$="--##.###"
7077 PRINT "DELTA LOG FO2               ";
7080 PRINT USING D$,R3,R4,R5
7090 PRINT "PRESSURE IN KILOBARS";
7095 PRINT TAB(51);L7
7500 T9=T9+TIMER/L9 @INCREMENTS THE PROGRAM TIME
7510 ZERO TIMER
7900 DISMISS
7999 GOTO 7000
8000 OPENW :2 "$QTO:0" @OPENS CHANNEL 2 FOR PRINTING---------------------------------------------------------------
8001 PNT 7 PNT :2 7 @WARNING BELL RINGS THRICE
8002 WAIT .4
8003 PNT 7 PNT :2 7
8004 WAIT .4
8005 PNT 7 PNT :2 7
8006 WAIT 30 @DELAY ASSURES THAT ALL INTERNAL VARIABLES ARE ASSIGNED VALUES
8010 PRINT :2 PTIME :2 PRINT :2 @COMMENCE PRINTING OUTPUT
8020 PRINT :2 "PGM TIME =";PRINT :2 T9;PRINT :2 "HRS"
8030 A$="--####.##"
8035 PRINT :2 "RUN TIME IN HOURS";
8040 PRINT USING:2"      "A$,T1,T2,T3,T4,T5,T6
8050 B$="--####.#"
8055 PRINT :2 "TEMP, CELSIUS   ";
8060 PRINT USING:2"       "B$,C1,C2,C3,C4,C5,C6
8070 C$="--####.###"
8075 PRINT :2 "OXYGEN FUGACITY IN LOG ATMOSPHERES            ";
8080 PRINT USING:2"     "C$,F3,F4,F5
8081 D$="--####.###"
8083 PRINT :2 "DELTA LOG ATM FO2 RELATIVE TO IRON-WUSTITE      ";
8084 PRINT USING:2"     "D$,R3,R4,R5
8085 PRINT :2 "PISTON CYLINDER PRESSURE IN KILOBARS"
8086 PRINT :2 TAB(96),L7
8090 PRINT :2
8100 CLOSE :2
8998 DISMISS
8999 GOTO 8000
```

APPENDIX III. Example of Printout During Run

```
CONSOLE INTERRUPT  AT LINE 994

(M) ALERT 3

(M) CON


04/27/81 12:16:55
PGM TIME =2.84088 HRS
RUN TIME IN HOURS            2.82          2.82          2.82          2.83          2.83          2.83
TEMP, CELSIUS              467.2         441.4        1122.3         952.3        1248.6          17.4
OXYGEN FUGACITY IN LOG ATMOSPHERES                      -13.222          0.011        -9.323
DELTA LOG ATM FO2 RELATIVE TO IRON-WUSTITE               -0.291         15.648         1.991
PISTON CYLINDER PRESSURE IN KILOBARS                                                                .0773544

CONSOLE INTERRUPT  AT LINE 994

(M) J8=920

(M) CON


04/27/81 12:19:10
PGM TIME =2.878 HRS
RUN TIME IN HOURS            2.82          2.82          2.87          2.86          2.85          2.87
TEMP, CELSIUS              467.2         441.4        1126.7         950.4        1248.4          18.6
OXYGEN FUGACITY IN LOG ATMOSPHERES                      -13.178          0.010        -9.315
DELTA LOG ATM FO2 RELATIVE TO IRON-WUSTITE               -0.308         15.682         1.990
PISTON CYLINDER PRESSURE IN KILOBARS                                                                .0758073


04/27/81 14:19:10
PGM TIME =4.85033 HRS
RUN TIME IN HOURS            4.79          4.79          4.85          4.83          4.83          4.84
TEMP, CELSIUS              801.7         712.8        1125.3         950.6        1248.3          20.1
OXYGEN FUGACITY IN LOG ATMOSPHERES                      -12.888          0.010        -9.316
DELTA LOG ATM FO2 RELATIVE TO IRON-WUSTITE                0.002         15.679         2.000
PISTON CYLINDER PRESSURE IN KILOBARS                                                                .0476503
```